

Elderly and Child Poverty Rates

Goal

Households with children and/or the elderly are usually at higher risk of poverty. The rising interest in analysing the incomes of these groups has increased the use of child and elderly poverty rates. (See, for example, the LIS key figures.)

Activity

Use data for Finland and the US in 2000. Calculate the Head Count Ratio and the Income Gap Ratio for the total population, the elderly, and for children.

Guidelines

- Prepare the data as you did in the previous exercise (drop observations with missing or zero *dpi*).
- All surveyed households and their members must be included in the estimates of the poverty line. After the (unique) poverty line has been calculated, only those households with either members under the age of 18 (for the child poverty figures) or over 64 (elderly figures) are included when computing the proportion of the population (subgroup) living in poverty.
- One way to consider a subgroup in your calculations is to change the weights. In this case, you can create two additional weights: one for households with children and one for households with elderly. These weights will be equal to 0 if there are no children/elderly in the household, but will be equal to the normal weight multiplied by the number of children/elderly (and not total number of household members!):

```
cwt = hweight * d27;
```

```
ewt = hweight * SUM(num6574, numge75);
```

- Try to use as much as possible the function **SUM** instead of using **+** as missing are not considered with the SUM whereas an error is generated while computation is done with **+**.
- For this exercise, construct the indicator three times: once for each group you want to examine. For each indicator, you will use the corrected weight.

Program

```
OPTIONS NONOTES NOSOURCE NOFMterr NODATE NONUMBER NOCENTER LABEL LS=max
PS=max ;
```

```
%MACRO avg;
  PROC MEANS DATA=povgrp MEAN ;
    VAR    poor eymed povlin gap;
    WEIGHT &poids    ;
  RUN;
%MEND avg;
%MACRO y(pi);
  DATA povgrp;
    SET &&pi.h (KEEP=hweight d4 d27 num6574 numge75 dpi);
    IF dpi in (. 0) THEN DELETE;
    ey  = dpi /SQRT(d4);
    wt  = hweight * d4 ;
    cwt = hweight * d27;
    ewt = hweight * SUM(num6574,numge75);
    n   = 1 ;
    IF  d27 > 0 THEN c = 1;
    IF (SUM(num6574,numge75)>0) THEN e = 1;
  RUN;
  PROC MEANS DATA=povgrp NOPRINT;
    VAR ey;
    WEIGHT wt ;
    OUTPUT OUT=temp MEDIAN=medey;
  RUN ;
  DATA _NULL_;
    SET temp;
    CALL SYMPUT("m",medey);
  RUN;
  DATA povgrp (KEEP=n c e eymed povlin gap poor wt cwt ewt) ;
    SET povgrp;
    poor = 0 ;
    eymed = &m ;
    povlin = &m * 0.5 ;
    IF ey < povlin THEN
      DO;
        gap = povlin-ey;
        poor = 1 ;
      END;
  RUN ;
  TITLE1 "COUNTRY : " &pi ;
  TITLE2 "***** NUMBER OF OBSERVATIONS *****" ;
  PROC FREQ DATA=povgrp ;
    TABLES n c e / MISSPRINT NOROW NOCOL NOCUM MISSING ;
  RUN ;
  TITLE2 "***** TOTAL POPULATION *****" ;
  %LET poids = wt ;
  %avg
  TITLE2 "***** CHILDREN *****" ;
  %LET poids = cwt ;
  %avg
  TITLE2 "***** EDERLY *****" ;
  %LET poids = ewt ;
```

```
%avg  
%MEND y ;  
%y(fi00)  
%y(us00)
```

Results

	# Observations	Poverty line	HCR	IGR
Finland 2000				
Total	<i>10 421</i>	<i>48,166.15</i>	<i>5.43%</i>	<i>25.57%</i>
with children	<i>3 893</i>	<i>48,166.15</i>	<i>2.97%</i>	<i>44.70%</i>
with elderly	<i>1 837</i>	<i>48,166.15</i>	<i>8.47%</i>	<i>12.20%</i>
US 2000				
Total	<i>49 351</i>	<i>12,046.99</i>	<i>17.05%</i>	<i>33.45%</i>
with children	<i>18 030</i>	<i>12,046.99</i>	<i>21.93%</i>	<i>34.13%</i>
with elderly	<i>11 634</i>	<i>12,046.99</i>	<i>24.72%</i>	<i>29.60%</i>

Comments

- As already mentioned above, the poverty line remains the same whether one examines the entire population, children, or the elderly but the households included in calculating the percentage below the threshold change with each subgroup. Therefore, keep in mind that the sample sizes of these subgroup rates are based on fractions of the entire sample and treat these figures with care when interpreting your results.